Osciladores lineares

Fernando Deeke Sasse

Departmento de Matemática

CCT - UDESC

Consideremos a equação que descreve um sistema forçado periodicamente, sem amortecimento:

```
> eq:=m*diff(x(t),t$2)+kappa*x(t)=sin(omega*t);
```

$$eq := m\left(\frac{d^2}{dt^2}\mathrm{x}(t)\right) + \kappa\,\mathrm{x}(t) = \sin(\omega\,t)$$

Solução geral da parte homogênea:

```
>
> eqh:=lhs(eq)=0;
```

$$eqh := m\left(\frac{d^2}{dt^2}\mathrm{x}(t)\right) + \kappa\,\mathrm{x}(t) = 0$$

```
> solh:=dsolve(eqh,x(t));
```

$$solh := \mathrm{x}(t) = \_C1\,\sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) + \_C2\,\cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right)$$

```
> xh:=rhs(solh);
```

$$xh := \_C1\,\sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) + \_C2\,\cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right)$$

Solução particular da eq. não homogênea:

```
>
> xp:=A*cos(omega*t)+B*sin(omega*t);
```

$$xp := A\cos(\omega\,t) + B\sin(\omega\,t)$$

```
> eq1:=subs(x(t)=xp,eq);
```

$$eq1 := m\left(\frac{\partial^2}{\partial t^2}(A\cos(\omega\,t) + B\sin(\omega\,t))\right) + \kappa\,(A\cos(\omega\,t) + B\sin(\omega\,t)) = \sin(\omega\,t)$$

```
> eq1;
```

$$m\,(-A\cos(\omega\,t)\,\omega^2 - B\sin(\omega\,t)\,\omega^2) + \kappa\,(A\cos(\omega\,t) + B\sin(\omega\,t)) = \sin(\omega\,t)$$

```
> ex1:=lhs(%)-rhs(%);
```

$$ex1 := m\,(-A\cos(\omega\,t)\,\omega^2 - B\sin(\omega\,t)\,\omega^2) + \kappa\,(A\cos(\omega\,t) + B\sin(\omega\,t)) - \sin(\omega\,t)$$

```
> E1:=coeff(ex1,cos(omega*t));
```

$$E1 := -m\,A\,\omega^2 + \kappa\,A$$

```
> E2:=coeff(ex1,sin(omega*t));
```

$$E2 := -m\,B\,\omega^2 + \kappa\,B - 1$$

```
> solp:=solve({E1=0,E2=0},{A,B});
```

$$solp := \{A = 0, B = -\frac{1}{m\,\omega^2 - \kappa}\}$$

Aqui temos que supor que $\omega \neq \sqrt{\frac{\kappa}{m}}$

```
> assign(solp);
> xp;
```

$$-\frac{\sin(\omega\,t)}{m\,\omega^2 - \kappa}$$

Solução geral:

```
>
> xg:=xh+xp;
```

$$xg := \_C1 \sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) + \_C2 \cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) - \frac{\sin(\omega\,t)}{m\,\omega^2 - \kappa}$$

Exemplo 1 . Sejam

```
>
> kappa:=1; m:=1; omega:=2;
```

$$\kappa := 1$$

$$m := 1$$

$$\omega := 2$$

e condições iniciais x(0) = 1, $x'(0) = -1$. Devemos então deteminar $\_C1$ e $\_C2$.

```
> E1:=subs(t=0,xg)=1;
```

$$E1 := \_C1 \sin(0) + \_C2 \cos(0) - \frac{1}{3} \sin(0) = 1$$

```
> E1;
```

$$\_C2 = 1$$

```
> E2:=subs(t=0,diff(xg,t))=-1;
```

$$E2 := \_C1 \cos(0) - \_C2 \sin(0) - \frac{2}{3} \cos(0) = -1$$

```
> E2;
```

$$\_C1 - \frac{2}{3} = -1$$

```
> sol2:=solve({E1,E2},{_C1,_C2});
```

$$sol2 := \{\_C2 = 1, \_C1 = \frac{-1}{3}\}$$

```
> assign(sol2);
> xg;
```

$$-\frac{1}{3}\sin(t)+\cos(t)-\frac{1}{3}\sin(2\,t)$$

```
> plot(xg,t=0..15);
```



Vamos encapsular os comandos anteriores para o oscilador forçado e amortecido.

```
> restart:
> m:=1.;g:=0.01;kappa:=1;w:=1;a:=-0.01;F:=sin(omega*t)*exp(a*t);x0:=
  1;v0:=1;
```

$$m := 1.$$

$$g := 0.01$$

$$\kappa := 1$$

$$w := 1$$

$$a := -0.01$$

$$F := \sin(\omega\, t)\, \mathbf{e}^{(-0.01\, t)}$$

$$x0 := 1$$

$$v0 := 1$$

```
> eq:=m*diff(x(t),t$2)+g*diff(x(t),t)+kappa*x(t)=F:
> eqh:=lhs(eq)=0:
> solh:=dsolve(eqh,x(t)):
> xh:=rhs(solh):
> xp:=(A*cos(omega*t)+B*sin(omega*t))*exp(a*t):
> eq1:=subs(x(t)=xp,eq):
> eq1:
> ex1:=simplify(lhs(%)-rhs(%)):
>
> E1:=subs(omega=w,coeff(ex1,cos(omega*t))):

> E2:=subs(omega=w,coeff(ex1,sin(omega*t))):
> omega:=w:
> solp:=solve({E1=0,E2=0},{A,B}):
> assign(solp):
> xg:=xh+xp:
> e1:=subs(t=0,xg)=x0:
> e2:=subs(t=0,diff(xg,t))=v0:
> sol2:=solve({e1,e2},{_C1,_C2}):
> assign(sol2):
> xg:=xg;
```

$$xg := 1.505018813\, \mathbf{e}^{\left(-\frac{t}{200}\right)} \sin\left(\frac{\sqrt{39999}\, t}{200}\right) - 99.\, \mathbf{e}^{\left(-\frac{t}{200}\right)} \cos\left(\frac{\sqrt{39999}\, t}{200}\right) + 100.\, \cos(t)\, \mathbf{e}^{(-0.01\, t)}$$

```
> plot(xg,t=0..1000, numpoints=2000);
```


20
10
0
-10
-20
200
400
600
800
1000
t

>
>
>
>